

1. Poids idéal

Calcul du poids idéal d'une personne à partir de plusieurs modèles médicaux.

Réf : <http://www.distrimed.com> et christianwtd.

1.1 Préliminaires

1.1.1 Expressions logiques

Une expression logique est un calcul (souvent une comparaison) dont le résultat est VRAI (valeur numérique 1) ou FAUX (valeur numérique 0).

Quelques exemples :

=A1= " f " (le premier signe = indique une formule, le second le signe de comparaison)

=A1= "F " (idem)

=A4 < 0

=A6 < moyenne(\$A\$6:\$A\$10)

pour tester la valeur numérique, il suffit par exemple d'ajouter 1 au résultat obtenu.

La fonction OU()

=OU(A1= "F " ; A1= "f ") (***)

1.1.2 Fonctions MAJUSCULE() , MINUSCULE() et NOMPROPRE()

On trouve ces fonctions dans la catégorie 'texte'.

Il n'est pas nécessaire de donner beaucoup d'explications...

=majuscule(A14)

=minuscule(A14)

=nompropre(A15)

Remarque

La formule (***) est équivalente à

=majuscule(A1)= "F " ou encore à =minuscule(A1)= "f "

1.2 Calculs

Pour la facilité de création et de lecture des résultats, on utilisera des *cellules nommées* :

B2 (sexe), B3 (taille), B4 (poids), B5 (age), B6 (morpho).



Disposition des données et des résultats :

	A	B	C	D	E
1	Calcul du poids idéal (www.distrimed.com)				
2	Sexe (M ou F)	M		masculin	
3	taille (cm)	170			
4	poids (kg)	67			
5	age (année)	45			
6	Morphologie (N,G,L)		N(normale) G(gracile) L(large)		
7	IMC (indice de masse corporelle)				
8	poids idéal				
9	formule de <u>Broca</u>	70			
10	formule de <u>Lorentz</u>	62			
11	formule de <u>Creff</u>				
12	Obésité				
13					

1.2.1 Encodage du sexe (B2)

Biologiquement, il n'y a que 2 cas possibles (mâle – femelle), ce qu'on représente habituellement en français par les lettres M (masculin) et F (féminin).

En toute logique, il faut donc tester si l'utilisateur a bien encodé l'une des 2 lettres, et rien d'autre.

La cellule D2 servira de contrôle, mais n'est pas indispensable pour les calculs, just for fun...

1.2.1.1 Solution 1

L'utilisateur a le choix de l'encodage, mais seules les lettres « M », « m », « F » et « f » sont autorisées; toute autre possibilité sera signalée par un message d'erreur; ce message d'erreur ne servira qu'à avertir l'utilisateur.

Raisonnement en B2 :

Si B2= « m » ou B2= « M »

alors « masculin »

sinon SI b2= « f » ou B2= « F »

alors « féminin »

sinon « erreur d'encodage »

Sous CALC, on utilisera les fonctions SI() et OU().

=SI(OU(B2= « M » ; B2= « m ») ; « masculin » ; SI(OU(B2= « F » ; B2= « f ») ; « féminin » ; « erreur d'encodage »))

... c'est plus difficile à lire qu'à écrire et à penser ...

1.2.1.2 Solution 2

Plus élégant, en utilisant la fonction MAJUSCULE() :



Si majuscule(B2)= «M»

alors « masculin »

sinon SI majuscule(B2)= «F»

alors « féminin »

sinon « erreur d'encodage »

=SI(majuscule(B2)= « M » ; « masculin » ; SI(majuscule(B2)= « F » ; « féminin » ; « erreur d'encodage »))


1.2.1.3 Solution 3

Le must.

On obligera l'utilisateur à encoder soit « M », soit « F » et rien d'autre, et donc plus d'erreur possible.

En B2 :

- **Données > Validité... > onglet Critères > Autoriser Liste**



en D2, on peut maintenant se limiter à

=si(B2= « M » ; « masculin » ; « féminin ») ou encore

=si(B2= « F » ; « féminin » ; « masculin »)



1.2.2 Formule de Broca


$$\text{poids idéal} = \text{taille} - 100$$

Cette formule est extrêmement basique.


Il faut tout d'abord que la taille soit >100, sinon le résultat est négatif.

Si la taille est égale à 101 (soit 1m01), le poids idéal serait de ... 1 kg ...

On peut éventuellement mettre une condition de validité sur la taille, par exemple >130 (voir corps médical...).



The screenshot shows the 'Validité' dialog box with the 'Critères' tab selected. The 'Autoriser' section has 'Nombre décimal' set to 2 and 'Autoriser les cellules vides' unchecked. The 'Données' section is set to 'supérieur à'. The 'Minimum' field contains the value '130'. At the bottom, there are buttons for 'OK', 'Annuler', 'Aide', and 'Rétablir'.



The screenshot shows the 'Validité' dialog box with the 'Message d'erreur' tab selected. The checkbox 'Afficher un message d'erreur en cas de saisie de valeurs incorrectes' is checked. In the 'Contenu' section, 'Action' is set to 'Information', 'Titre' is empty, and 'Message d'erreur' contains the text 'Il faut un nombre > 130'. At the bottom, there are buttons for 'OK', 'Annuler', 'Aide', and 'Rétablir'.

Formule en B9 : =taille - 100



1.2.3 Formule de Lorentz

Cette formule fait intervenir le sexe de la personne.

Homme : $taille - 100 - \left(\frac{taille - 150}{4}\right)$ Femme : $taille - 100 - \left(\frac{taille - 150}{2,5}\right)$
--

Raisonnement en B10:

Si sexe= « M »

Alors $taille - 100 - (taille - 150) / 4$

Sinon $taille - 100 - (taille - 150) / 2,5$

Formule en B10 :

=si(sexe = « M » ; $taille - 100 - (taille - 150) / 4$; $taille - 100 - (taille - 150) / 2,5$)

Remarque

Dans les 2 cas de l'alternative, on fait presque le même calcul, il n'y a que le dénominateur qui change.

Nouveau raisonnement en B10:

$taille - 100 - (taille - 150) /$

Si sexe= « M »

Alors 4

Sinon 2,5

Formule en B10 :

= $taille - 100 - (taille - 150) / \text{si}(\text{sexe} = \text{« M »} ; 4 ; 2,5)$

1.2.4 Formule de Creff

Cette formule fait intervenir la notion de morphologie (normale, gracile, large).

Normale : $(taille - 100 + \frac{age}{10}) * 0,9$ Gracile : $(taille - 100 + \frac{age}{10}) * 0,81$ Large : $(taille - 100 + \frac{age}{10}) * 0,99$

La cellule B6 contiendra l'une des 3 lettres N, G, et L (données > validité...)

En B11, on pourrait utiliser une formule du même genre que pour la formule de Lorentz :

=si(morpho = « N » ;) ou bien, plus court,

= $(taille - 100 + age / 10) * \text{si}(\text{morpho} = \text{« N »} ; 0,9 ; \text{si}(\text{morpho} = \text{« G »} ; 0,81 ; \dots))$



Une autre solution :

Une expression logique a pour valeur numérique 1 si elle est vraie, 0 si elle est fausse.

L'expression

$(\text{morpho} = \ll \text{N} \gg) * 0,9$ vaudra donc 0,9 lorsque morpho= « N » et 0 dans les autres cas.

$(\text{morpho} = \ll \text{G} \gg) * 0,81$ vaudra donc 0,81 lorsque morpho= « G » et 0 dans les autres cas.

$(\text{morpho} = \ll \text{L} \gg) * 0,99$ vaudra donc 0,99 lorsque morpho= « L » et 0 dans les autres cas.

Si on fait la somme des expressions, à savoir

$$=(\text{morpho} = \ll \text{N} \gg) * 0,9 + (\text{morpho} = \ll \text{G} \gg) * 0,81 + (\text{morpho} = \ll \text{L} \gg) * 0,99$$

on obtient :

0,9 lorsque morpho= « N »

0,81 lorsque morpho= « G »

0,99 lorsque morpho= « L »

à savoir le coefficient multiplicatif de $(\text{taille} - 100 + \text{age}/10)$.

Si on appelle « coeff_creff » une cellule contenant la somme ci-dessus, la formule en B11 sera :

$$= \text{coeff_creff} * (\text{taille} - 100 + \text{age}/10)$$

Masquer les cellules de travail.

La cellule coeff_creff ne doit pas être visible (c'est une cellule de travail); pour la masquer, il suffit de choisir comme couleur de police la même que la couleur de fond.

1.2.5 Indice de masse corporelle (IMC)

Cet indice donne des indications sur l'obésité.

$$IMC = \frac{\text{poids}}{\text{taille}^2} \quad \text{où la taille doit être exprimée en mètre!}$$

Formule en B7 :

$$= \text{poids} / (\text{taille}/100)^2$$



1.2.6 Interprétation de l'indice de masse corporelle

Le tableau suivant donne l'interprétation à donner :

IMC	Interprétation
[18,5 ; 25[Corpulence idéale
[25 ; 30[Surcharge pondérale
[30 ; 35[Obésité modérée
[35 ; 40[Obésité sévère
[40 ;	Obésité très sévère

Ce qu'il faut calculer dans la cellule B12.

L'IMC comprise entre inclus et 25 exclus s'exprime par l'expression

$$IMC \geq 18,5 \text{ et } IMC < 25$$

qui se traduit par la formule :

$$= ET (IMC \geq 18,5 ; IMC < 25)$$

et il en faut 5 du même genre.... avec des SI imbriqués, c'est tout à fait possible, mais il y a une fonction CHOISIR() qui rendra cela plus clair.

Syntaxe :

CHOISIR(valeur ; calcul si valeur=1 ; calcul si valeur=2;; calcul si valeur=n)

On va donc s'arranger pour que l'expression

$ET (IMC \geq 18,5 ; IMC < 25)$ vaut 1

$ET (IMC \geq 25 ; IMC < 30)$ vaut 2

$ET (IMC \geq 30 ; IMC < 35)$ vaut 3

$ET (IMC \geq 35 ; IMC < 40)$ vaut 4

$(IMC \geq 40)$ vaut 5

chacune de ces expressions logiques vaut 1 si elle est VRAIE; il faut donc multiplier chacune d'entre elle par le coefficient correspondant, et additionner le tout.

Le plus simple est de créer une cellule de travail (nommée *valeur* par exemple) qui contiendra la fameuse somme :

$$=ET (IMC \geq 18,5 ; IMC < 25) * 1 + ET (IMC \geq 25 ; IMC < 30) * 2 + ET (IMC \geq 30 ; IMC < 35) * 3 + ET (IMC \geq 35 ; IMC < 40) * 4 + (IMC \geq 40) * 5$$

B12 contiendra alors

=CHOISIR(valeur ; « corpulence idéale » ; « surcharge pondérale » ; « obésité modérée » ; « obésité sévère » ; « obésité très sévère »)

Bon appétit !

